

[Description](#)
[Suboptions](#)
[Also see](#)

[Quick start](#)
[Remarks and examples](#)

[Menu](#)
[Stored results](#)

[Syntax](#)
[Methods and formulas](#)

Description

`margins` with the `pwcompare` option performs pairwise comparisons of margins. `margins, pwcompare` extends the capabilities of `pwcompare` to any of the nonlinear responses, predictive margins, or other margins that can be estimated by `margins`.

Quick start

All pairwise comparisons of the predictive margins for `a` after `logit y a##b x1`
`margins a, pwcompare`

Same as above, but report both tests and confidence intervals for differences in predictive margins
`margins a, pwcompare(effects)`

Same as above, but adjust p -values and confidence intervals for multiple comparisons using Bonferroni's method
`margins a, pwcompare(effects) mcompare(bonferroni)`

Report predictive margins for the levels of `a`, and group those that are not significantly different
`margins a, pwcompare(groups)`

All pairwise comparisons of the predictive margins for combinations of levels of `a` and `b`
`margins a#b, pwcompare`

Menu

Statistics > Postestimation

Syntax

```
margins [marginlist] [if] [in] [weight] [, pwcompare margins_options]
```

```
margins [marginlist] [if] [in] [weight] [, pwcompare(suboptions) margins_options]
```

where *marginlist* is a list of factor variables or interactions that appear in the current estimation results. The variables may be typed with or without the `i.` prefix, and you may use any factor-variable syntax:

```
. margins i.sex i.group i.sex#i.group, pwcompare
. margins sex group sex#i.group, pwcompare
. margins sex##group, pwcompare
```

<i>suboptions</i>	Description
-------------------	-------------

Pairwise comparisons

<code>cieffects</code>	show effects table with confidence intervals; the default
<code>pveffects</code>	show effects table with <i>p</i> -values
<code>effects</code>	show effects table with confidence intervals and <i>p</i> -values
<code>cimargins</code>	show table of margins and confidence intervals
<code>groups</code>	show table of margins and group codes
<code>sort</code>	sort the margins or contrasts in each term

`collect` is allowed; see [U] 11.1.10 Prefix commands.

`fweights`, `awweights`, `iweights`, and `pweights` are allowed; see [U] 11.1.6 weight.

Suboptions

Pairwise comparisons

`cieffects` specifies that a table of the pairwise comparisons with their standard errors and confidence intervals be reported. This is the default.

`pveffects` specifies that a table of the pairwise comparisons with their standard errors, test statistics, and *p*-values be reported.

`effects` specifies that a table of the pairwise comparisons with their standard errors, test statistics, *p*-values, and confidence intervals be reported.

`cimargins` specifies that a table of the margins with their standard errors and confidence intervals be reported.

`groups` specifies that a table of the margins with their standard errors and group codes be reported. Margins with the same letter in the group code are not significantly different at the specified significance level.

`sort` specifies that the reported tables be sorted on the margins or contrasts in each term.

[stata.com](http://www.stata.com)

Remarks and examples

You should be familiar with the concepts and syntax of both `margins` and `pwcompare` before using the `pwcompare` option of `margins`. These remarks build on those in [R] [margins](#) and [R] [pwcompare](#).

`margins` can perform pairwise comparisons of any of the margins that it estimates.

We begin by fitting a logistic regression model using the NHANES II dataset, ignoring the complex survey nature of the data. Our dependent variable is `highbp`, an indicator for whether a person has high blood pressure. We fit an interacted model including two factor variables representing the region of the country as well as the continuous covariate `bmi`.

```
. use https://www.stata-press.com/data/r18/nhanes2
. logistic highbp region#c.bmi
  (output omitted)
```

By default, `margins` will compute the predictive margins of the probability of a positive outcome for each of the terms in `marginlist` after logistic regression. We will `margin` on `region` so that `margins` will estimate the average predicted probabilities of having high blood pressure conditional on being in each of the four regions and unconditional on BMI. We can specify the `pwcompare` option to obtain all possible pairwise comparisons of these predictive margins:

```
. margins region, pwcompare
Pairwise comparisons of predictive margins          Number of obs = 10,351
Model VCE: OIM
Expression: Pr(highbp), predict()
```

	Delta-method		Unadjusted	
	Contrast	std. err.	[95% conf. interval]	
region				
MW vs NE	-.0377194	.0133571	-.0638987	-.01154
S vs NE	-.0156843	.0133986	-.041945	.0105764
W vs NE	-.006873	.0136595	-.0336451	.019899
S vs MW	.0220351	.0124564	-.0023789	.0464492
W vs MW	.0308463	.0127366	.0058831	.0558096
W vs S	.0088112	.0127801	-.0162373	.0338598

This table gives each of the pairwise differences with confidence intervals. We can see that the confidence interval in the row labeled `MW vs NE` does not include 0. At the 5% level, the predictive margins for the first and second regions, the Northeast and the Midwest, are significantly different. The same is true of the second and fourth regions, the Midwest and the West. With many pairwise comparisons, output in this format can be difficult to sort through. We can organize it by adding the `group` suboption:

4 margins, pwcompare — Pairwise comparisons of margins

```
. margins region, pwcompare(group)
Pairwise comparisons of predictive margins      Number of obs = 10,351
Model VCE: OIM
Expression: Pr(highbp), predict()
```

region	Delta-method		Unadjusted groups
	Margin	std. err.	
NE	.4388358	.010069	B
MW	.4011164	.0087764	A
S	.4231516	.0088395	AB
W	.4319628	.0092301	B

Note: Margins sharing a letter in the group label are not significantly different at the 5% level.

The group output includes the predictive margins for each region and letters denoting margins that are not significantly different from one another. In this case, the Northeast (NE), South (S), and West (W) regions have the letter B in the “Unadjusted Groups” column. The letter B indicates that the average predicted probability for the Northeast region is not significantly different from the average predicted probabilities for the South and West regions at the 5% significance level. The Midwest (MW) region does not share a letter with the Northeast region nor the West region, which indicates that the average predicted probability for the Midwest region is significantly different for each of the other two regions at our 5% level.

We can also include the `mcompare(bonferroni)` option to perform tests using Bonferroni’s method to account for making multiple comparisons.

```
. margins region, pwcompare(group) mcompare(bonferroni)
Pairwise comparisons of predictive margins      Number of obs = 10,351
Model VCE: OIM
Expression: Pr(highbp), predict()
```

region	Number of comparisons
region	6

region	Delta-method Bonferroni		groups
	Margin	std. err.	
NE	.4388358	.010069	B
MW	.4011164	.0087764	A
S	.4231516	.0088395	AB
W	.4319628	.0092301	AB

Note: Margins sharing a letter in the group label are not significantly different at the 5% level.

We now see the letter A on the row corresponding to the West region. At the 5% level and with Bonferroni’s adjustment, the predictive margins for the probability in the Midwest and West regions are not significantly different.

Stored results

`margins`, `pwcompare` stores the following additional results in `r()`:

Scalars

`r(k_terms)` number of terms participating in pairwise comparisons

Macros

`r(cmd)` `pwcompare`
`r(cmd2)` `margins`
`r(group#)` group code for the #th margin in `r(b)`
`r(mcmethod_vs)` *method* from `mcompare()`
`r(mctitle_vs)` title for *method* from `mcompare()`
`r(mcadjustall_vs)` `adjustall` or empty

Matrices

`r(b)` margin estimates
`r(V)` variance-covariance matrix of the margin estimates
`r(b_vs)` margin difference estimates
`r(V_vs)` variance-covariance matrix of the margin difference estimates
`r(error_vs)` margin difference estimability codes;
 0 means estimable,
 8 means not estimable
`r(table_vs)` matrix containing the margin differences with their standard errors, test statistics, *p*-values, and confidence intervals
`r(L)` matrix that produces the margin differences

`margins`, `pwcompare` with the `post` option also stores the following additional results in `e()`:

Scalars

`e(k_terms)` number of terms participating in pairwise comparisons

Macros

`e(cmd)` `pwcompare`
`e(cmd2)` `margins`

Matrices

`e(b)` margin estimates
`e(V)` variance-covariance matrix of the margin estimates
`e(b_vs)` margin difference estimates
`e(V_vs)` variance-covariance matrix of the margin difference estimates
`e(error_vs)` margin difference estimability codes;
 0 means estimable,
 8 means not estimable
`e(L)` matrix that produces the margin differences

Methods and formulas

See *Methods and formulas* in [\[R\] margins](#) and *Methods and formulas* in [\[R\] pwcompare](#).

Also see

- [\[R\] contrast](#) — Contrasts and linear hypothesis tests after estimation
- [\[R\] margins](#) — Marginal means, predictive margins, and marginal effects
- [\[R\] margins, contrast](#) — Contrasts of margins
- [\[R\] margins postestimation](#) — Postestimation tools for margins
- [\[R\] pwcompare](#) — Pairwise comparisons

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).